
G04 – Analysis of Variance

G04DBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G04DBF computes simultaneous confidence intervals for the differences between means. It is intended
for use after G04BBF or G04BCF.

2 Specification

SUBROUTINE G04DBF(TYPE, NT, TMEAN, RDF, C, LDC, CLEVEL, CIL, CIU,
1 ISIG, IFAIL)
INTEGER NT, LDC, ISIG(NT*(NT-1)/2), IFAIL
real TMEAN(NT), RDF, C(LDC,NT), CLEVEL,
1 CIL(NT*(NT-1)/2), CIU(NT*(NT-1)/2)
CHARACTER∗1 TYPE

3 Description

In the computation of analysis of a designed experiment the first stage is to compute the basic analysis
of variance table, the estimate of the error variance (the residual or error mean square), σ̂2, the residual
degress of freedom, ν, and the (variance ratio) F -statistic for the t treatments. The second stage of
the analysis is to compare the treatment means. If the treatments have no structure, for example the
treatments are different varieties, rather than being structured, for example a set of different temperatures,
then a multiple comparison procedure can be used.

A multiple comparison procedure looks at all possible pairs of means and either computes confidence
intervals for the difference in means or performs a suitable test on the difference. If there are t treatments
then there are t(t − 1)/2 comparisons to be considered. In tests the type 1 error or significance level is
the probability that the result is considered to be significant when there is no difference in the means.
If the usual t-test is used with, say, a five percent significance level then the type 1 error for all k =
t(t − 1)/2 tests will be much higher. If the tests were independent then if each test is carried out at the
100α percent level then the overall type 1 error would be α∗ = 1− (1−α)k � kα. In order to provide an
overall protection the individual tests, or confidence intervals, would have to be carried out at a value of
α such that α∗ is the required significance level, e.g., five percent.

The 100(1− α) percent confidence interval for the difference in two treatment means, τ̂i and τ̂j is given
by

(τ̂i − τ̂j)± T ∗
(α,ν,t)se(τ̂i − τ̂j),

where se() denotes the standard error of the difference in means and T ∗
(α,ν,t) is an appropriate percentage

point from a distribution. There are several possible choices for T ∗
(α,ν,t). These are:

(a) 1
2q(1−α,ν,t), the studentised range statistic, see G01FMF. It is the appropriate statistic to compare
the largest mean with the smallest mean. This is known as Tukey–Kramer method.

(b) t(α/k,ν), this is the Bonferroni method.
(c) t(α0,ν), where α0 = 1− (1− α)1/k, this is known as the Dunn–Sidak method.
(d) t(α,ν), this is known as Fisher’s LSD (least significant difference) method. It should only be used

if the overall F -test is significant, the number of treatment comparisons is small and were planned
before the analysis.

(e)
√
(k − 1)F1−α,k−1,ν where F1−α,k−1,ν is the deviate corresponding to a lower tail probability of

1− α from an F -distribution with k − 1 and ν degrees of freedom. This is Scheffe’s method.

In cases (b), (c) and (d), t(α,ν) denotes the α two-tail significance level for the Student’s t-distribution
with ν degrees of freedom, see G01FBF.

The Scheffe method is the most conservative, followed closely by the Dunn–Sidak and Tukey–Kramer
methods.

To compute a test for the difference between two means the statistic,

τ̂i − τ̂j

se(τ̂i − τ̂j)

is compared with the appropriate value of T ∗
(α,ν,t).

4 References

of Statistical Sciences 5 Wiley, New York

[2] Kotz S and Johnson N L (ed.) (1985) Multiple comparison Encyclopedia of Statistical Sciences 5
Wiley, New York

[3] Winer B J (1970) Statistical Principles in Experimental Design McGraw–Hill

5 Parameters

1: TYPE — CHARACTER*1 Input

On entry: indicates which method is to be used.

If TYPE = ’T’, the Tukey–Kramer method is used.
If TYPE = ’B’, the Bonferroni method is used.
If TYPE = ’D’, the Dunn–Sidak method is used.
If TYPE = ’L’, the Fisher LSD method is used.
If TYPE = ’S’, the Scheffe’s method is used.

Constraint: TYPE = ’T’,’B’,’D’, ’L’ or ’S’.

2: NT — INTEGER Input

On entry: the number of treatment means, t.

Constraint: NT ≥ 2.

3: TMEAN(NT) — real array Input

On entry: the treatment means, τ̂i, i = 1, 2, . . . , t.

4: RDF — real Input

On entry: the residual degrees of freedom, ν.

Constraint: RDF ≥ 1.0.

5: C(LDC,NT) — real array Input

On entry: the strictly lower triangular part of C must contain the standard errors of the differences
between the means as returned by G04BBF and G04BCF. That is C(i, j), i > j, contains the
standard error of the difference between the ith and jth mean in TMEAN.

Constraint: C(i, j) > 0.0, i = 2, 3, . . . , t; j = 1, 2, . . . , i− 1.

6: LDC — INTEGER Input

On entry: the first dimension of the array C as declared in the (sub)program from which G04DBF
is called.

Constraint: LDC ≥ NT.

7: CLEVEL — real Input

On entry: the required confidence level for the computed intervals, (1− α).

Constraint: 0.0 < CLEVEL < 1.0.

8: CIL(NT*(NT−1)/2) — real array Output

On exit: the ((i − 1)(i − 2)/2 + j)th element contains the lower limit to the confidence interval for
the difference between ith and jth means in TMEAN, i = 2, 3, . . . , t; j = 1, 2, . . . , i − 1.

9: CIU(NT*(NT−1)/2) — real array Output

On exit: the ((i − 1)(i − 2)/2 + j)th element contains the upper limit to the confidence interval for
the difference between ith and jth means in TMEAN, i = 2, 3, . . . , t; j = 1, 2, . . . , i − 1.

10: ISIG(NT*(NT−1)/2) — INTEGER array Output

On exit: the ((i − 1)(i − 2)/2 + j)th element indicates if the difference between ith and jth means
in TMEAN is significant, i = 2, 3, . . . , t; j = 1, 2, . . . , i − 1. If the difference is significant then the
returned value is 1; otherwise the returned value is 0.

11: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Errors and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

On entry, NT < 2,

or LDC < NT,

or RDF < 1.0,

or CLEVEL ≤ 0.0,

or CLEVEL ≥ 1.0,

or TYPE �= ’T’, ’B’, ’D’, ’L’ or ’S’.

IFAIL = 2

On entry, C(i, j) ≤ 0.0 for some i, j, i = 2, 3, . . . , t; j = 1, 2, . . . , i − 1.

IFAIL = 3

There has been a failure in the computation of the studentized range statistic. This is an unlikely
error. Try using a small value of CLEVEL.

7 Accuracy

For the accuracy of the percentage point statistics see G01FMF and G01FBF.

8 Further Comments

If the treatments have a structure then the use of linear contrasts as computed by G04DAF may be more
appropriate.

An alternative approach to one used in this routine is the sequential testing of the Student–Newman–
Keuls procedure. This, in effect, uses the Tukey–Kramer method but first ordering the treatment means
and examining only subsets of the treatment means in which the largest and smallest are significantly
different. At each stage the third parameter of the Studentised range statistic is the number of means in
the subset rather than the total number of means.

9 Example

In the example taken from Winer [3] a completely randomised design with unequal treatment replication
is analysed using G04BBF and then confidence intervals are computed by G04DBF using the Tukey–
Kramer method.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G04DBF Example Program Text
* Mark 17 Release. NAG Copyright 1995.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, NTMAX, NBMAX
PARAMETER (NMAX=26,NTMAX=4,NBMAX=1)

* .. Local Scalars ..
real CLEVEL, GMEAN, RDF, TOL
INTEGER I, IFAIL, IJ, IRDF, J, N, NBLOCK, NT
CHARACTER TYPE

* .. Local Arrays ..
real BMEAN(NBMAX), C(NTMAX,NTMAX),

+ CIL(NTMAX*(NTMAX-1)/2), CIU(NTMAX*(NTMAX-1)/2),
+ EF(NTMAX), R(NMAX), TABLE(4,5), TMEAN(NTMAX),
+ WK(NTMAX*NTMAX+NTMAX), Y(NMAX)
INTEGER IREP(NTMAX), ISIG(NTMAX*(NTMAX-1)/2), IT(NMAX)
CHARACTER STAR(2)

* .. External Subroutines ..
EXTERNAL G04BBF, G04DBF

* .. Executable Statements ..
WRITE (NOUT,*) ’G04DBF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, NT
IF (N.LE.NMAX .AND. NT.LE.NTMAX) THEN

READ (NIN,*) (Y(I),I=1,N)
READ (NIN,*) (IT(I),I=1,N)
TOL = 0.000005e0
IRDF = 0
NBLOCK = 1
IFAIL = -1
CALL G04BBF(N,Y,NBLOCK,NT,IT,GMEAN,BMEAN,TMEAN,TABLE,4,C,NTMAX,

+ IREP,R,EF,TOL,IRDF,WK,IFAIL)
WRITE (NOUT,*)
WRITE (NOUT,*) ’ ANOVA table’
WRITE (NOUT,*)
WRITE (NOUT,*)

+ ’ Source df SS MS F’,
+ ’ Prob’

WRITE (NOUT,*)
WRITE (NOUT,99998) ’ Treatments’, (TABLE(2,J),J=1,5)
WRITE (NOUT,99998) ’ Residual ’, (TABLE(3,J),J=1,3)
WRITE (NOUT,99998) ’ Total ’, (TABLE(4,J),J=1,2)
WRITE (NOUT,*)
WRITE (NOUT,*) ’ Treatment means’
WRITE (NOUT,*)

WRITE (NOUT,99999) (TMEAN(J),J=1,NT)
WRITE (NOUT,*)
WRITE (NOUT,*) ’ Simultaneous Confidence Intervals’
WRITE (NOUT,*)
RDF = TABLE(3,1)
READ (NIN,*) TYPE, CLEVEL

*
CALL G04DBF(TYPE,NT,TMEAN,RDF,C,NTMAX,CLEVEL,CIL,CIU,ISIG,

+ IFAIL)
*

STAR(2) = ’*’
STAR(1) = ’ ’
IJ = 0
DO 40 I = 1, NT

DO 20 J = 1, I - 1
IJ = IJ + 1
WRITE (NOUT,99997) I, J, CIL(IJ), CIU(IJ),

+ STAR(ISIG(IJ)+1)
20 CONTINUE
40 CONTINUE

END IF
STOP

*
99999 FORMAT (10F8.3)
99998 FORMAT (A,3X,F3.0,2X,2(F10.1,2X),F10.3,2X,F9.4)
99997 FORMAT (2X,2I2,3X,2(F10.3,3X),A)

END

9.2 Program Data

G04DBF Example Program Data

26 4

3 2 4 3 1 5
7 8 4 10 6
3 2 1 2 4 2 3 1
10 12 8 5 12 10 9

1 1 1 1 1 1
2 2 2 2 2
3 3 3 3 3 3 3 3
4 4 4 4 4 4 4

’T’ .95

[NP3390/19/pdf] G04DBF.5



G04DBF G04 – Analysis of Variance

9.3 Program Results

G04DBF Example Program Results

ANOVA table

Source df SS MS F Prob

Treatments 3. 239.9 80.0 24.029 0.0000
Residual 22. 73.2 3.3
Total 25. 313.1

Treatment means

3.000 7.000 2.250 9.429

Simultaneous Confidence Intervals

2 1 0.933 7.067 *
3 1 -3.486 1.986
3 2 -7.638 -1.862 *
4 1 3.610 9.247 *
4 2 -0.538 5.395
4 3 4.557 9.800 *

G04DBF.6 (last) [NP3390/19/pdf]


